Упражнение 8
Функции

С конструкцията deffunction могат да се дефинират нови функции директно в CLIPS. Дефинираните чрез deffunction функции са еквивалентни на другите вградени в CLIPS функции. Единствената разлика между потребителски дефинираните външни функции и тези дефинирани чрез deffunction е че deffunction са написани на CLIPS и се интерпретират, а потребителските външни функции са написани на външен език като C и директно се изпълняват от CLIPS. Deffunction дава възможност и за добавяне на нови функции без да е необходимо да се прекомпилира CLIPS. Deffunction се състои от пет елемента: 1) име, 2) незадължителен коментар, 3) списък от нула или повече задължителни параметри, 4) незадължителен wildcard параметър за работа с променлив брой аргументи и 5) последователност от действия или изрази, които се изпълняват поред когато се извика функцията.

Синтаксиса на конструкцията deffunction е:

(deffunction <name> [<comment>]

    (<regular-parameter>* [<wildcard-parameter>])

    <action>*)

<regular-parameter>  ::= <single-field-variable>

<wildcard-parameter> ::= <multifield-variable>

Deffunction трябва да има уникално име, различно от всички други системни функции и генерични функции. В частност, deffunction не може да се предефинира (без значение от броя на параметрите) като системна функция. Deffunction трябва да се декларира преди да се извика от друга deffunction, метод на генерична функция, object message-handler, правило или top level. Единственото изключение са deffunction с пряка рекурсия.

Deffunction могат да приемат точно или най-малко определен брой аргументи, в зависимост от това дали е използван wildcard параметър или не. Задължителните параметри определят минилания брой аргументи, които трябва да се подадат на deffunction. Всеки от тези параметриможе да се използва като нормална single-field променлива в тялото на функцията. Ако е зададен wildcard параметър, deffunction трябва да получи точния брой аргументи определен от задължителните параметри. Всички аргументи на функцията  които не отговарят на задължителните параметри се групират в multifield стойност която може да се използва чрез wildcard параметъра.

Пример:

CLIPS> (clear)

CLIPS>

(deffunction print-args (?a ?b $?c)

(
printout t ?a " " ?b " and " (length ?c) " extras: " ?c crlf))

CLIPS> (print-args 1 2)

1 2 and 0 extras: ()

CLIPS> (print-args a b c d)

a b and 2 extras: (c d)

CLIPS>

При обръщение към функция, действията от тялото и се изпълняват последователно. Връщаната стойност на функцията е оценката на последното действие. Ако функцията не изпълнява действия, тя връща стойност FALSE. Ако при изпълнението на функцията възникне грешка, неизпълнените действия се отхвърлят (не се изпълняват) и функцията връща FALSE.

Deffunction могат да изпълняват пряка или косвена рекурсия. Пряката рекурсия се реализира чрез извикване на функцията от нейното собствено тяло.

Пример:

(deffunction factorial (?a)


(if (or (not (integerp ?a)) (< ?a 0)) then



(printout t "Factorial Error!" crlf)


else



(if (= ?a 0) then




1



else



(* ?a (factorial (- ?a 1))))


)

)

Взаимната рекурсия между две функции изисва използването на специална forward декларация в една от функциите. Forward декларацията е просто декларация на функция без тяло. В следващия пример  функцията foo е декларирана като forward, така че тя може да се извика от функцията bar. След това се извършва редефиниране на функцията foo с тяло в което се извършва извикване на функцията bar.

Пример:

(deffunction foo ())

(deffunction bar ()


(foo))

(deffunction foo ()


(bar))

Трябва да се внимава при използването на рекурсивните функции, тъй като твърде многото нива на рекурсия могат да доведат до препълване на стековата памет (особено при PC машини).

defgeneric/defmethod
С конструкциите defgeneric/defmethod могатда се напишат нови генерични функции директно в CLIPS. Генеричните функции са подобни на deffunction тъй като могатда се използват за дефиниране на нов процедурен код директно в CLIPS и могат да се извикват както всяка друга функция. Все пак генеричните функции са много по-мощни тъй като те могат да извършват различнидействия в зависимост от типовете (или класовете) и броя аргументи. Например оператора '+' може да се дефинира да изпълнява конкатенация на низове но все пак да изпълнява и аритметичната операция събиране за числа. Една генерична функция се състои от заглавна част (подобна на forward декларация) и нула или повече компоненти наречени методи, като всеки метод обработва различна комбинация от аргументи за генеричната функция. Заглавната част на функцията може да бъде декларирана или явно от потребителя или неявно чрез дефиницията на поне един метод. Конструкцията defdgeneric се използва за задаване на заглавна част на генерична функция, а defmethod се използва за задаване на методи на генерични функции. Генерична функция която има повече от един метод се нарича предефинирана (overloaded).

(defgeneric <name> [<comment>])

(defmethod <name> [<index>] [<comment>]

    (<parameter-restriction>* [<wildcard-parameter>])

    <action>*)

<parameter-restriction> ::= <single-field-variable> |

                            (<single-field-variable> <type>* [<query>])

<wildcard-parameter>    ::= <multifield-variable>

<type>                  ::= <class-name>

<query>                 ::= <global-variable> | <function-call>

Генеричните функции могат да използват системни функции и потребителски дефинирани външни функции като неявни методи. Например предефинирането на оператора "+" за низове се състои от два метода: 1)неявен който е системна функция изпълняваща събиране на числа, и 2)явен (потребителски дефиниран) изършващ конкатенация на низове. Deffunction все пак може и да не съставен от методи на генерични функции. Deffunction е предоставен така че да могат да се добавят нови основни функции директно в CLIPS без да е необходимо предефиниране. Например генерична функция която има само един метод който ограничава само броят на аргументите е еквивалентна на deffunction.

В повечето случаи методите на генерични функции не се извикват директно (за тази цел може да се използва call-specific-method). CLIPS разпознава че функционалното извикване е генерично и използва аргументите на генеричното извикване за да намери и изпълни подходящ метод. Този процес се нарича generic dispatch.

Използване на методи

Повечето ООП системи поддържат процедурно поведение на обекти или чрез предаване на съобщения (напр. Smalltalk) или чрез генерични функции (напр. CLOS). CLIPS поддържа и двата механизма, въпреки че генеричните функции не се поддържат строго от COOL. Една генерична функция може да разпита класовете за техните аргументи, но все пак трябва да използва съобщения в телата на методите си за да работи с аргументите си които са инстанци на потребителски дефинирани класове. Фактът че CLIPS поддържа и двата механизъма води до объркване в технологиите. В ООП системите поддържащи само предаване на съобщения, терминът метод се използва за означаване на различни имплементации на съобщение за различни класове. За преодоляване на това объркване, терминът message-handler се използва за представянена метода в контекста на съобщения. Така в CLIPS, message-handlers означават различните имплементации на съобщение за различни класове, а методи (methods) означават различни имплементации на генерични функции (generic function) за различните множества от ограничения върху аргументите.

Недостатъчци в производителността при генеричните функции

Извикването на генерична функция е с по-голяма изчислителна цена от извикването на системна функция, потребителски дефинирана външна функция или deffunction. Това е така, тъй като CLIPS трябва първо да провери аргументите на функцията за да определи кой метод да използва. Ограниченията върху производителността могат да са около 15% - 20%. Така генеричните функции не трябва да се използват за процедури при които времето е критичен фактор, като например процедуриизвиквани в while цикъл, ако изобщо е възможно. Генеричните функции освен това трябва да имат поне два метода. Deffunctions или потребителски дефинираните външни функции трябва да се използват когато не се изисква предефиниране. Системна или потребителски дефинирана външна функция която не е предефинирана (overloaded) ще се изпълни достатъчно бързо, тъй като generic dispatch не е необходим.

Ако конструкция използваща системна или потребителски дефинирана външна функция се зарежда преди генеричната функция използваща тази функциякато неявен метод, всички извиквания към тази функция от тази конструкция ще прескочат generic dispatch. Например ако генерична функция която предефинира (overload) оператора "+" е дефинирана след правило което използва оператор "+", това правило винаги ще извиква състемната функция "+" директно. Все пак, подобни правила дефинирани след генерична функция ще използват generic dispatch.

Дефиниране на нова генерична функция

Генеричните функции се състоят от заглавна част (хедър) (подобна на forward декларация) и нула или повече методи. Заглавната част на генеричната функция може да бъде декларирана или явно от потребителя или неявно чрез дефиниране на поне един метод. Методът се състои от шест елемента: 1) име (което определя към коя генерична функция принадлежи метода), 2) незадължителен индекс, 3) незадължителен коментар, 4) множество от ограничения върху параметрите, 5) ограничения върху незадължителния wildcard параметър за използване на променлив брой аргументи и 6) последователност от действия или изрази, които се изпълняват поред при извикване на метода. Ограниченията върху параметрите се използват от генеричната функция за извършване на generic dispatch за определяне на приложимостта на метода към множеството от аргументи при действителното извикване на генеричната функция. Конструкцията defgeneric  се използва за задаване на заглавната част на генеричната функция, а конструкцията defmethod за всеки метод на генеричната функция.

(defgeneric <name> [<comment>])

(defmethod <name> [<index>] [<comment>]


(<parameter-restriction>* [<wildcard-parameter-restriction>])


<action>*)

<parameter-restriction> ::=


<single-field-variable> |


(<single-field-variable> <type>* [<query>])

<wildcard-parameter-restriction> ::=


<multifield-variable> |


(<multifield-variable> <type>* [<query>])

<type> ::= <class-name>

<query> ::= <global-variable> |


<function-call>

Генеричната функция преди да се извика от метод на друга генерична функция, deffunction, object message-handler, правило или от най-високото ниво трябва да се декларира чрез заглавна част и метод. Еединственото изключение е при генеричните функции с пряка рекурсия.

Заглавни части на генерични функции

Генеричната функция се идентифицира уникално по име. За да се извърши обръщение към генерична функция от други конструкции преди някой от нейните методи да се декларира, е необходимо явно да се зададе заглавна част. В противен случай заглавната част се създава неявно от декларацията на първия метод. Например две генерични функции чиито методивзаимно се извикват (взаимно рекурсивни генерични функции) изискват явни заглавнаи части.

Индекси на методи

Методите уникално се идентифицират чрез име и индекс или чрез име и ограничения върху параметрите. На всеки метод от генерична функция му се назначава уникален индекс в групата от всички методи за тази генерична функция. Така ако се дефинира нов метод със същото име и същите ограничения върху параметрите като предишния, CLIPS автоматично ще премахне стария метод. Все пак ако има някакви разлики в ограниченията към параметрите, това ще предизвика дефинирането на нов метод в допълнение към стария. За да се замести стария метод с този който има различни ограничения върху параметрите е необходимо индекса на стария метод да се зададе явно в дефиницията на новия. Ограниченията върху параметрите на новия метод обаче не трябвада съответстват на ограниченията на друг метод с различен индекс. Ако не се зададе индекс, CLIPS назначава неизползван от който и да е метод на генеричната функция индекс (преди или в момента на задаването). Индекса назначен от CLIPS може да се види чрез командата list-defmethod.

Ограничения върху параметрите на метод

Всеки параметър наметод може да се дефинира така че да има някакви по-сложни ограничения или да няма никакви. Ограниченията към параметрите се прилагат към аргументите на генерична функция по време на изпълнение (run-time) за определяне дали даден метод може да приеме тези аргументи. Даден параметър може да има два типа ограничения: тип и заявка. Ограниченията по тип определят класовете на аргументите които могат да бъдат приети за даден параметър. Ограниченията заявки са булеви изрази които трябва да се удовлетворяват за даден аргумент, за да бъде приемлив. Сложността на ограниченията на параметрите директно оказва влияние върху скоростта на извършване на generic dispatch.

Ако параметърът няма ограничения, то това означава че методът приемя какъвто и да е аргумент на тази позиция. Всеки метод на генерична функция обаче трябва да има ограничения върху параметрите си, което го различава от всички други методи, така че generic dispatch да може да определи кой от тях да извика по време на изпълнение. Ако няма приложими методи за дадено извикване на генерична функция, CLIPS ще генерира грешка.

Ограниченията по типове дават възможност на потребителя да определи списък от типове (или класове), някой от които трябва да съответства (или да е суперклас) на класа на аргумента на генеричната функция. Ако не е инсталиран COOL в текущата CLIPS конфигурация, единствените възможни типове (или класове) са: OBJECT, PRIMITIVE, LEXEME, SYMBOL, STRING, NUMBER, INTEGER, FLOAT, MULTIFIELD, FACT-ADDRESS и EXTERNAL-ADDRESS. С COOL могат да се използват също INSTANCE, INSTANCE-ADDRESS, INSTANCE-NAME, USER, INITIAL-OBJECT и някой потребителски дефинирани класове. Генерични функции които използват само първата група от типове в техните методи ще работят по същия начин независимо дали е инсталиран COOL или не. Класовете в ограничеията за типове трябва да са дефинирани, тъй като те се използват за определяне на приоритите между методите на генеричните функции. Не е разрешено използването на излишни класове в списъка от ограничаващи класове. например следните ограничения върху типа на параметър на метод е излишен, тъй като INTEGER е подклас на NUMBER.

(defmethod foo ((?a INTEGER NUMBER)))

Ако съществува ограничение за тип и ако то се удовлетворява, то ако съществува ограничение заявка тогава и тя ще се приложи. Ограничението заявка трябва да бъде или глобална променлива или функционално извикване. CLIPS изчислява израза и ако резултатът му е различен от FALSE ограничението се удовлетворява. Тъй като ограниченията върху параметрите се проверяват отляво надясно, заявките за множествени параметри трябва да бъдат включени с най-десния параметър. Това гарантира че всички ограничения върху типовете на параметрите вече са удовлетворени. Например следния метод забавя изчислението на ограничението заявка докато класовете на двата аргумента не се проверят.

(defmethod foo ((?a INTEGER) (?b INTEGER (> ?a ?b))))

Ако аргумента премине всички тези тестове, той се счита за приемлив за метода. Ако всички аргументи на генерични функции са приемливи за ограниченията на метода, самия метод се приема за приложим за множеството аргументи. Ако повече от един метод е приложим към множеството от аргументи, generic dispatch трябва да определи редът им и изпълнението на първия от тях в този ред. За тази цел се използва приоритет на методите който ще бъде разгледан по-късно.

Пример

В следния пример първото извикване на генеричната функция "+" изпълнява системния оператор "+", неявен метод за събиране на числа. Второто извикване изпълнява явние метод за конкатенация на низове, тъй като има два аргумента като и двата са низове. Третото извикване генерира грешка тъй като явния метод за конкатенация на низове приема само два аргумента, а неявния метод за събиране на числа не приема низове.

CLIPS> (clear)

CLIPS>

(defmethod + ((?a STRING) (?b STRING))


(str-cat ?a ?b))

CLIPS> (+ 1 2)

3

CLIPS> (+ "foo" "bar")

"foobar"

CLIPS> (+ "foo" "bar" "woz")

[GENRCEXE1] No applicable methods for +.

FALSE
Незадължителен (wildcard) параметър на метод

Даден метод може да приема точно или поне зададен брой аргументи в зависимост от това дали е използван незадължителен параметър или не. Задължителните параметри задават минималния брой аргументи които трябва да се предадат на метода. Всеки от тези параметри може да се използва като нормална single-field променлива в тялото на метода. Ако има незадължителен параметър, метода може да приема произволен брой аргументи по-голям или равен на минималния. Ако няма незадължителен параметър, методът може да приема точен брой аргументи определен, чрез задължителните параметри. Аргументите на метод, които не съответстват на задължителния параметър, могат да се групират в multifield променлива която може да се използва, чрез незадължителния параметъра в тялото на метода. Към този незадължителен параметър могат да се прилагат стандартните CLIPS функции за работа с множетсвени стойности като напр. length$ и expand$.

Ако се предават множествени стойности като допълнителни аргументи, всички те ще бъдат обединени в една multifield стойност отнесена към незадължителния параметър. Това е така, тъй като CLIPS не поддържа вложени multifield стойности.

Ограниченията по тип и заявка могат да се прилагат към аргументите групирани в незадължителния параметър подобно на редеовните параметри. такива ограничения приложени към всяко отделно поле на резултатната multifield стойност (не целия multifield). Изразите в незадължителен параметър обаче може да се използва в заявката. Допълнително в ограниченията заявки може да се използва специална променлива върху незадължителния параметър за позоваване на отделните аргументи групирани в незадължителния параметър: ?current-argument. Тази променлива има област на действие само в заявката и няма значение в тялото на метода. например за създаване на версия на оператора "+" който действа различно за суми от всички четни целочислени стойности:

CLIPS>

(defmethod +


(($?any INTEGER (evenp ?current-argument)))


(div (call-next-method) 2))

CLIPS> (+ 1 2)

3

CLIPS> (+ 4 6 4)

7

CLIPS>

Важно е да се подчертае че ограниченията по тип и заявка за незадължителния параметър се прилагат към всеки от елементите групирани в незадължителния параметър. Така в следния пример функциите > и length$ в действителност се извикват три пъти, тъй като има три аргумента:

CLIPS> (defmethod foo (($?any (> (length$ ?any) 2))) yes)

CLIPS> (foo 1 red 3)

yes

CLIPS>

В допълнение, ограничението заявка никога няма да се провери ако няма аргументи в зоната на незадължителния параметър. Например предишния метод би бил приложим към извикването на генеричната функция без аргументи, тий като ограничението заявка не се изчислява:

CLIPS> (foo)

yes
CLIPS>

Обикновено ограниченията заявка приложени към целия незадължителен параметър проверяват cardinality (броят аргументи подадени към метода). В случаи като този при които типа не е подходящ за проверката, ограниченията заявка могат да се поделят със задължителен параметър, за да се подобри производителността. Така предишния метод може да се оптимизира по следния начин:

CLIPS> (clear)

CLIPS> (defmethod foo ((?arg (> (length$ ?any) 1)) $?any) yes)

CLIPS> (foo)

[GENRCEXE1] No applicable methods for foo.

FALSE
CLIPS>

Този подход не трябва да се използва ако типовете на аргументите групирани от незадължителния параметър трябва да се проверяват преди безопасно проверяване на ограниченията по заявка.

Generic dispatch
При извикване на генерична функция, CLIPS избира метод за тази генерична функция с най-висок приоритет за който ограниченията върху параметрите се удовлетворяват от аргументите. Методът се изпълнява и връща стойността си като стойност на генеричната функция. Целият този процес се нарича generic dispatch. По-долу има обобщаваща диаграма.

Приложимост на методи. Обощение

Явно зададен (потребителки дефиниран) метод е приложим към извикване към генерична функция ако са изпълнени следните три условия: 1) нейното име съотвтства на това на генеричната функция, 2) тя приема поне толкова аргументи колкото са подадени към генеричната функция и 3) всеки аргумент на генеричната функция удовлетворява съответните ограничения върху параметъра (ако съществуват) в метода.

Ограниченията на метод се проверяват отляво надясно. Ако едно ограничение не се удовлетворява, методът се прекратява и останалите ограничения (ако има такива) не се проверяват.

Ако стандартна системна функция на CLIPS се предефинира, CLIPS формира дефиниция на неявен метод отговарящ на тази системна функция. Този неявен метод се наследява от низа на ограниченията на аргументите за външното извикване на DefineFunction2 дефиницията на тази функция към CLIPS (виж  Advanced Programming Guide). Този низ може да се достъпва с функцията get-function-restriction. Спецификацията на този неявен метод може да се види с функциите list-defmethods или get-method-restrictions. Методът който CLIPS ще формира за системна функция може да се наследи от потребителя от БНФ от документацията. Например

(+ <number> <number>+)

ше създаде следния мотод за функцията '+':

(defmethod + ((?first NUMBER) (?second NUMBER) ($?rest NUMBER)) ...)

Дефиницията на метода се използва за определяне на приложимостта и приоритета на системната функция към извикването на генеричната функция.

Следните системни функции не могат да се предефинират, тъй като CLIPS ще генерира грешка.

active-duplicate-instance delayed-do-for-all-instances message-modify-instance

active-initialize-instance do-for-all-instances modify

active-make-instance do-for-instance modify-instance

active-message-duplicate-instance duplicate next-handlerp

active-message-modify-instance duplicate-instance next-methodp

active-modify-instance expand$ object-pattern-match-delay

any-instancep find-all-instances override-next-handler

assert find-instance override-next-method

bind if progn

break make-instance progn$

call-next-handler initialize-instance return

call-next-method loop-for-count switch

call-specific-method message-duplicate-instance while

Приоритет на методите

Когато два или повече метода са приложими към дадено извикване на генерична функция, CLIPS трябва да избере един от тях който е с най-висок приоритет за изпълнение. Приоритетът на методите се определя при дефинирането на методите. Може да се използва функцията list-defmethods за определяне на приоритета на методите за генерична функция.

Приоритета между два метода се определя чрез сравняване на ограниченичта върху техните параметри. Най-общо метода с най-специфични ограничения върху параметрите има най-висок приоритет. Например метод който изисква за аргумент целочислена стойност (integer) ще има по-висок приоритет от метод който изисква число (number). Точните правила за определяне на предимството между два метода са дадени по-долу. Взет е резултата от първото правило което получава предимство.

1) Ограниченията върху параметрите на два метода се сравняват позиционно отляво надясно. С други думи, първото ограничение върху параметър върху първия метод съответства на първото ограничение във втория метод и т.н. Сравненията между тези двойки от ограничения върху параметрите от два метода определят цялостното предимство измежду два метода. Взема се резултатът от сравнението на първата двойка ограничения върху параметрите. Следнващите правила се прилагат към двойка от параметри; взема се резултатът от първото правило което може да установи предимството.


1а) Един задължителен параметър има предимство пред незадължителен параметър.


1б) Най-специфичните ограничения върху тип за даден параметър има приоритет. Клас е по-специфичен от някой от неговите суперкласове.


1с) Параметър с ограничение заявка има приоритет пред този който няма такова.

2) Методът с по-голям брой задължителни параметри има предимство.

3) Метод без незадължителен параметър има предимство над този който има такъв.

4) Метод дефиниран преди друг има предимство.

Ако има множество класове в дадено единично ограничение, определянето на специфичността е малко по-сложно. Тъй като при дефиниране на нов метод се извършва определяне на всички приоритети, а действителния клас на аргумента на генерична функция не е известен до момента на изпълнение, правилата трябва да се използват произволно (но детерминирано) за определяне на предимството между два списъка от класове. Двата списъка от класове се проверяват по двойки отляво надясно, т.е. двойката от първите класове от двата списъка, двойките от вторите класове от двата списъка и т.н. Първата двойка съдържаща клас и негов суперклас определя предимството. Списъкът от класове съдържащ подклас има приоритет. Ако няма двойка класове определяща предимството, тогава се избира най-късия списък от класове за по-приоритетен. В противен случай списъка от класове не определя предимство от ограниченията на параметрите.

Пример:

; The system operator '+' is an implicit method ; #1

; Its definition provided by the system is:

; (defmethod + ((?a NUMBER) (?b NUMBER) ($?rest NUMBER)))

(defmethod + ((?a NUMBER) (?b INTEGER))) ; #2

(defmethod + ((?a INTEGER) (?b INTEGER))) ; #3

(defmethod + ((?a INTEGER) (?b NUMBER))) ; #4

(defmethod + ((?a NUMBER) (?b NUMBER)($?rest PRIMITIVE))) ; #5

(defmethod + ((?a NUMBER)(?b INTEGER (> ?b 2)))) ; #6

(defmethod + ((?a INTEGER (> ?a 2))(?b INTEGER (> ?b 3)))) ; #7

(defmethod + ((?a INTEGER (> ?a 2))(?b NUMBER))) ; #8

Приоритетът ще бъде: #7,#8,#3,#4,#6,#2,#1,#5. Методите могат веднага да бъдат разделени в три групи по намаляващ приоритет според ограниченията им върху първия параметър: А) методи които имат ограничение заявка и ограничение по тип INTEGER (#7, #8), В) методи които имат ограничение по тип INTEGER (#3, #4), С) методи които имат ограничение по тип NUMBER (#1, #2, #5, #6). Група А има предимство над група В тъй като параметрите с ограничение заявка имат приоритет пред тези без такова. Група В има предимство пред група С тъйкато INTEGER е подклас на NUMBER. Така редът е: (#7,#8),(#3,#4),(#1,#2,#5,#6). Подреждането на методите в определено множество в скобите все още не е окончателно.

Следващата стъпка в определянето на приоритета между тези методи разглежда техните ограничения върху втория параметър. #7 има приоритет над #8, тъй като INTEGER е подклас на NUMBER. #3 има приоритет над #4 тъй като INTEGER е подклас на NUMBER. #6 и #2 имат приоритет над #1 и #5, тъй като INTEGER е подклас на NUMBER. #6 има приоритет над #2 тъй като има ограничение заявка, а #2 няма. Така се получава реда: #7,#8,#3,#4,#6,#2,(#1,#5).

Ограничението върху wildcard аргумента прави #1 (неявен метод от системна функция) по-приоритетен от #5, тъй като NUMBER е подкалс на PRIMITIVE. Това дава крайното подреждане: #7,#8,#3,#4,#6,#2,#1,#5.

Пример 2

(defmethod foo ((?a NUMBER STRING))) ; #1

(defmethod foo ((?a INTEGER LEXEME))) ; #2

Подреждането ще бъде: #2, #1. Въпреки че STRING е подклас на LEXEME, подреждането все пак е #2, #1, тъй като INTEGER е подклас на NUMBER и NUMBER/INTEGER е най-лявата двойка в списъка от класове.

Пример 3

(defmethod foo ((?a MULTIFIELD STRING))) ; #1

(defmethod foo ((?a LEXEME))) ; #2

Приоритетът ще бъде #2, #1 тъй като класовете от първата двойка в граниченията по тип (MULTIFIELD/LEXEME) нямат връзка и #2 има по-малко класове в списъка си от класове.

Пример 4

(defmethod foo ((?a INTEGER LEXEME))) ; #1

(defmethod foo ((?a STRING NUMBER))) ; #2

Двете двойки от класове (INTEGER/STRING и LEXEME/NUMBER) нямат връзка по между си и списъците от класове са с еднаква дължина. така приоритета се взема от реда на дефиниране: #1, #2.

Shadowed методи

Ако един метод трябва да се извика от друг метод за да се изпълни, първата функция или метод се нарича shadowed от втория метод. Обикновено към определено извикване на генерична функция са приложими само един метод или системна функция. Ако има повече от един приложим метод, generic dispatch ще изпълни този който има най-висок приоритет. Възможността generic dispatch автоматично да управлява методите по този начин се нарича декларативна техника за декларациите на ограниченията на метод определящи кой метод да се изпълнява при определени ситуации. Могат обаче да се използват и функциите call-next-method и override-next-method което дава възможност на даден метод да изпълни shadow метод. Това се нарича императивна техника, тъй като изпълнението на самия метод играе ролята на generic dispatch. Това не се препоръчва освен при крайни обстоятелства. В повечето ситуации само една част от кода трябва да се изпълни за определено множество от аргументи. Друга императивна техника е да се използва функцията call-specific-method за предефиниране на приоритет на метод.

Грешки при изпълнение на методи

Ако по време на изпълнение на метод на генерична функция възникне грешка, всички неизпълнени действия се отхвърлят и генеричната функция връща FALSE. Отсъствието на приложим метод за множеството от аргументи на генерична функция се разглежда като грешка при изпълнение на метод.

Връщане на стойност от генерична функция

Върнатата стойност от генерична функция е върната стойност на приложимия метод с няй-висок приоритет (предимство). Всеки приложим метод който се изпълнява може да избере дали да игнорира или да прихване върнатата стойност на метод на който той се явява shadow метод.

Върнатата стойност на даден метод е последното действие изчислено от този метод, т.е. последната оценка на изпълненото действие.

Цикли

loop-for-count
Функцията дава възможност за извършване на прост итеративен цикъл.

Синтаксис

(loop-for-count <range-spec> [do] <action>*)

<range-spec>  ::= <end-index> |

(<loop-variable> [<start-index> <end-index>])

<start-index> ::= <integer-expression>

<end-index>   ::= <integer-expression>

Действието се изпълнява зададеният брой пъти определен чрез <rangespec>. Ако <startindex> не е зададен се приема за единица. Ако <startindex> е по-малък от <endindex> тялото на цикъла на се изпълнява. Целочислената стойност  на текущата итерация може да се провери в тялото на цикъла чрез цикличната променлива ако има такава. Функциите break и return могат да се използват за извършване на преждевременно прекъсване на цикъла. Върнатата стойност на тази функция е FALSE освен ако return не се използва за прекъсване на цикъла. Променливите от външния цикъл могат да се използват във вътрешен цикъл, но цикличната променлива (ако има такава) маскира останалите външни променливи със същото име. Циклите могат да бъдат вложени.

Пример:

CLIPS> (loop-for-count 2 (printout t "Hello world" crlf))

Hello world

Hello world

FALSE

CLIPS>

(loop-for-count (?cnt1 2 4) do 

(loop-for-count (?cnt2 1 3) do

(printout t ?cnt1 " " ?cnt2 crlf)))

2 1

2 2

2 3

3 1

3 2

3 3

4 1

4 2

4 3

FALSE
CLIPS>

While
Функцията while дава възможност за извършване на прости цикли. Нейното използване е подобно на функцията if.

Синтаксис

(while <expression> [do]

   <action>*)

В while могат да се използват всички предикатни функции. В тялото на while блока могат да се използват множество от действия, включително if…then…else или допълнителна while структура. Проверката на условието се извършва преди първото изпълнение на цикъла. Действията в while цикъла се изпълняват докато <expression> получи оценка FALSE. While може да включва и незадължителен символ do след условието и преди първото действие. За преждевременно прекратяване на цикъла могат да се използват функциите break и return. Връщаната стойност от функцията е FALSE освен ако не се използва функцията return за прекратяване на цикъла.

Пример:

(defrule open-valves

   (valves-open-through ?v)

   =>

   (while (> ?v 0)

      (printout t "Valve " ?v " is open" crlf)

      (bind ?v (- ?v 1))))

===========================================================================

Задачи за самостоятелна работа:

1. Да се напише генерична функция съставена от два метода. Първият да приема множествена стойност от тип integer и връща минималната стойност. Вторият да приема множествена стойност от тип float и връща максималната стойност.
